```
> restart:
```

TOR0SPIN.mws
02/14/99 -- R. M. Kiehn

A Beltrami field that yields no spin but finite torsion, and is a wave solution with no currents or charge densities - if the classic dispersion relation is satisfied. Poincare invariants are zero! In other words, this solution exhibits propagating torsion with NO spin.

Maxell-Faraday formulas from a 1-form of Action on 4D space time.
Maxwell Ampere equations assuming Lorentz vacuum constitutie equations
D = epsilon E   B = mu H
Computes Torsion vector = ExA+Bphi, AdotB
computes Spin Vector = A x H + Dphi, AdotD
First Poincare invariant is F^G -A^J = (BdotH-DdotE) -(AdotJ-rhi.phi)
Second Poincare Invariant = F^F = -2EdotB
THE GENERAL PROCEDURE IS:
Given A,phi. compute E,B.   compute D,H.   compute J,rho, compute poincare invariants,
compute torsion current and spin current.

Fundamental Lagrangian is a 3-form. L = A^G +iA^F/Z(hall)

two DIFFERENT topological invariants are possible: torsion and spin (helicity and spin)
Invariance requires that divergence of Spin = 0 and divergence of Torsion = 0.

d(A^F) = F^F   ---- (torsion)

d(A^G)=F^G-A^J   ----(spin)

The Invariance conditions are distinct.

Spin can be conserved, Torsion can be conserved. But the are independent ideas.

```
>  with(liesymm):with(linalg):with(plots):
```

```
> setup(x,y,z,t);
```

$$[x, y, z, t]$$

```
> defform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,m=const);
```

$$\mathrm{defform}(x = 0, y = 0, z = 0, t = 0, a = const, b = const, c = const, k = const, \mu = const, m = const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [\,\mathrm{d}(x),\, \mathrm{d}(y),\, \mathrm{d}(z),\, \mathrm{d}(t)\,]$$

Specify the four functions that are the covariant components of the Action 1-form.

```
> A1:=Ax(x,y,z,t);
> A2:=Ay(x,y,z,t);
```

$$A1 := \mathrm{Ax}(x, y, z, t)$$

$$A2 := \mathrm{Ay}(x, y, z, t)$$

```
> A3:=Az(x,y,z,t);
```

$$A3 := \mathrm{Az}(x, y, z, t)$$

```
> A4:=phi(x,y,z,t);
```

$$A4 := \phi(x, y, z, t)$$

```
> r:=(x^2+y^2+z^2-(c*t)^2)^(1/2);
```

$$r := \sqrt{x^2 + y^2 + z^2 - c^2\, t^2}$$

Skip the next line for abstract formulas, or enter your own formulas for the vector and scalar potentials.
DO NOT CHANGE ANY OF THE OTHER EQUATIONS
****

```
> A1:=cos(k*z-omega*t);A2:=sin(k*z-omega*t);A3:=0*alpha*c*t;A4:=0*beta*c*sin(k*(z-c*t));
```

$$A1 := \cos(k\,z - \omega\,t)$$

$$A2 := \sin(k\,z - \omega\,t)$$

$$A3 := 0$$

$$A4 := 0$$

****

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t);
```

$$Action := \cos(k\,z - \omega\,t)\,\mathrm{d}(x) + \sin(k\,z - \omega\,t)\,\mathrm{d}(y)$$

The 2-form of Field Intensities obtained from the 1-form of Action.

```
> F:=wcollect(d(Action));
```

$$F := -\sin(k\,z - \omega\,t)\,k\,(\mathrm{d}(z)\ \&\wedge\ \mathrm{d}(x)) + \sin(k\,z - \omega\,t)\,\omega\,(\mathrm{d}(t)\ \&\wedge\ \mathrm{d}(x)) + \cos(k\,z - \omega\,t)\,k\,(\mathrm{d}(z)\ \&\wedge\ \mathrm{d}(y)) \\ - \cos(k\,z - \omega\,t)\,\omega\,(\mathrm{d}(t)\ \&\wedge\ \mathrm{d}(y))$$

The components of the Vector potential in engineering format.

```
> A:=evalm([A1,A2,A3]);
```

$$A := [\,\cos(k\,z - \omega\,t),\, \sin(k\,z - \omega\,t),\, 0\,]$$

Magnetic field intensity (Beltrami field)

```
> B:=curl(A,[x,y,z]);
```

$$B := [\,-\cos(k\,z - \omega\,t)\,k,\, -\sin(k\,z - \omega\,t)\,k,\, 0\,]$$

The Electric Field Intensity

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)];
```

$$E := [\,-\sin(k\,z - \omega\,t)\,\omega,\, \cos(k\,z - \omega\,t)\,\omega,\, 0\,]$$

Topological Parity (Second Poncare invariant) depends on topology of field intensities.

```
> EdotB:=factor(innerprod(E,B));
```

$$EdotB := 0$$

The 4 components of the topological Torsion current.

```
> ExA:=simplify(crossprod(E,A));Bphi:=[B[1]*A4,B[2]*A4,(B[3]*A4)];
```

$$ExA := [\,0,\, 0,\, -\omega\,]$$

$$Bphi := [\,0,\, 0,\, 0\,]$$

```
> TORS:=simplify(evalm(ExA+A4*B));
```

$$TORS := [0, 0, -\omega]$$

```
> AdotB:=simplify(factor(inner(A,B)));
```

$$AdotB := -k$$

AdotB is the classic helicity -- This is not a scalar or a pseudo scalar, but it is the fourth component of a third rank covariant tensor.

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$$TORSION := [0, 0, -\omega, -k]$$

Divergence of the Torsion current.

```
> DIVT:=factor(diverge(TORSION,[x,y,z,t]));
```

$$DIVT := 0$$

The Next step is to compute the Field Excitations assuming a Lorentz vacuum constitutive relation, B = mu H and D = e E. Once having D and H , then compute charge current density from the Maxwell Ampere equations.

The second Poincare Invariant is the related to the difference between the magnetic and the electric energy densities (the lagrangian) minus the interaction between the 4 current and the 4 potential.

First compute the Lagrangian, and along the way the components of the Poynting vector. The results are to within a scalar factor:

```
> EdotE:=simplify(innerprod(E,E)):BdotB:=innerprod(B,B):ExB:=crossprod(E,B):Lag:=simplify(factor(-EdotE+c^2*BdotB));POYX:=factor(ExB[1]);POYY:=factor(ExB[2]);POYZ:=simplify(factor(ExB[3]));Hamiltonian:=simplify(factor(EdotE+c^2*BdotB));
>
```

$$Lag := -\omega^2 + k^2 c^2$$

$$POYX := 0$$

$$POYY := 0$$

$$POYZ := k\,\omega$$

$$Hamiltonian := \omega^2 + k^2 c^2$$

Poynting vector in the z direction is finite.

Now the components of the 4 current:

```
> J:= evalm(curl(B,[x,y,z])*c^2-[diff(E[1],t),diff(E[2],t),diff(E[3],t)]):
> J1:=(factor(J[1]));J2:=factor(J[2]);J3:=factor(J[3]);rho:=factor(diverge(E,[x,y,z]));
```

$$J1 := \cos(k\,z - \omega\,t)\,(k\,c - \omega)\,(k\,c + \omega)$$

$$J2 := \sin(k\,z - \omega\,t)\,(k\,c - \omega)\,(k\,c + \omega)$$

$$J3 := 0$$

$$\rho := 0$$

NO charge current density (a wave solution) when $k\,c - \omega$=0.

Next compute the Interaction between the 4-current and the 4 potential (the interaction vanishes when wave equation is satisfied.)

```
> INTERACTION:=simplify(factor(innerprod(A,J)-A4*rho));
>
```

$$INTERACTION := -\omega^2 + k^2 c^2$$

Check to see if the fields satisfy the wave equation (requires that the charge current density be zero)

```
> CCB:=curl(curl(B,[x,y,z]),[x,y,z]):DBT:=[factor(diff(B[1],t)),factor(diff(B[2],t)),factor(diff(B[3],t))]:DDBT:=factor(diff(DBT[1],t));
```

$$DDBT := \cos(k\,z - \omega\,t)\,\omega^2\,k$$

```
> factor(CCB[1]);
```

$$-\cos(k\,z - \omega\,t)\,k^3$$

One factor of the vector wave equation. It should be zero if the solutions are wave functions.

```
> factor(-CCB[1]-DDBT/c^2);
```

$$\frac{\cos(k\,z - \omega\,t)\,k\,(k\,c - \omega)\,(k\,c + \omega)}{c^2}$$

Subject to the dispersion relation, the solution is a propagating wave solution.
Now evaluate the spin 3-form A^G
This object is not the same as the torsion 3-form, A^F
Often a duality arguement is made to say that they are the same. But this defeats the Lorentz vacuum constitutive constraint. REsults are to within a factor.

```
> Spin:=evalm(c^2*crossprod(A,B)+A4*E):spin4:=factor(innerprod(A,E));spin1:=factor(Spin[1]);spin2:=factor(Spin[2]);spin3:=factor(Spin[3]);
>
```

$$spin4 := 0$$

$$spin1 := 0$$

$$spin2 := 0$$

$$spin3 := 0$$

The spin vanishes, but the torsion is finite.
Must fix units later.

```
> SPINVECTOR:=[spin1,spin2,spin3,spin4];TORSIONVECTOR:=TORSION;DIV4SPIN:=factor(diverge(SPINVECTOR,[x,y,z,t]));DIV4TORS:=DIVT;EDOTB:=EdotB;Interaction:=INTERACTION;Lagrangian:=Lag;Poincare1:=Lagrangian-Interaction;Poincare2=EDOTB;
```

$$SPINVECTOR := [0, 0, 0, 0]$$

$$TORSIONVECTOR := [0, 0, -\omega, -k]$$

$$DIV4SPIN := 0$$

$$DIV4TORS := 0$$

$$EDOTB := 0$$

$$Interaction := -\omega^2 + k^2\,c^2$$

$$Lagrangian := -\omega^2 + k^2\,c^2$$

$$Poincare1 := 0$$

$$Poincare2 = 0$$

Therefore if the Spin 3-form has a zero divergence, it has an evolutionary invariant period integral!!!!.
But if the Spin 3-form is non-zero, then its divergence is equal to the First Poincare Invariant =F^G - A^J.

```
> CHECKSUM:=simplify(factor(DIV4SPIN-Lagrangian+Interaction));
```

$$CHECKSUM := 0$$

The CHECKSUM must vanish if the computations are correct.

****

Next compute the product of the Torsion vector with the 1-form of vector potential. It should vanish if the computations are correct..

```
> IEIRR:=factor(TORSION[1]*A[1]+TORSION[2]*A[2]+TORSION[3]*A[3]-AdotB*A4);
```

$$IEIRR := 0$$

Compute the Lorentz force in the direction of the Torsion vector. It should be proportional to the original 1-form, if the computations are correct.

```
> FLIRR:=evalm((E*AdotB+crossprod(TORS,B))):FLTORS:=[factor(FLIRR[1]),factor(FLIRR[2]),factor(FLIRR[3])];PLIRR:= simplify(factor(innerprod(TORS,E)));
```

$$FLTORS := [0, 0, 0]$$

$$PLIRR := 0$$

It is apparent that evolution in the direction of the torsion vector is conformal with the dissipation factor proportional to EdotB

>

The Torsion vector is a characteristic vector on the 4D space, when EdotB vanishes

>

>